ICE05 Argonne, July 19, 2005

Solution of Leader-Follower Games

Sven Leyffer leyffer@mcs.anl.gov

Mathematics & Computer Science Division,
Argonne National Laboratory

NASH & STACKELBERG GAMES

Recap: Nash Games

Nash Game: non-cooperative equilibrium of several players

$$z_i^* \in \left\{ \begin{array}{ll} \underset{z_i}{\text{argmin}} & b_i(\hat{z}) \\ \text{subject to} & c_i(z_i) \geq 0 \end{array} \right. \qquad \text{player } i$$

- $\hat{z} = (z_1^*, \ldots, z_{i-1}^*, z_i, z_{i+1}^*, \ldots, z_l^*)$
- All players are equal

Recap: Nash Games

Introduce slacks s, and form optimality conditions ...

$$\begin{aligned} \nabla b(z) - \nabla c(z)\lambda &= 0 \\ s - c(z) &= 0 \\ 0 \le \lambda \perp s &\ge 0 \end{aligned}$$

where

- $b(z) = (b_1(z), \ldots, b_k(z))$ & $c(z) = (c_1(z), \ldots, c_k(z))$
- $\perp$ means $\lambda^T s = 0$, either $\lambda_i > 0$ or $s_i > 0$

$$\left.\begin{array}{l} y = (z, \lambda, s) \\ h = \ldots \end{array}\right\} \quad \text{... becomes ...} \quad \left\{\begin{array}{l} h(y) = 0 \\ 0 \le y_1 \perp y_2 \ge 0 \end{array}\right.$$

- Nonlinear complementarity problem (NCP)
- Robust large scale solvers exist: `PATH`

Stackelberg Games

Single dominant player (leader) & Nash followers

$$\left\{ \begin{array}{ll} \underset{x \geq 0, y}{\text{minimize}} & f(x,y) \\ \text{subject to} & h(x,y) = 0 \\ & 0 \leq y_1 \perp y_2 \geq 0 \end{array} \right.$$



Nash game ($h(x,y) = 0$) parameterized in leader's variables x

Mathematical Program with Equilibrium Constraints (MPEC)

Mathematical Program with Equilibrium Constraints

Mathematical Program with Equilibrium Constraints (MPEC)

$$\left\{ \begin{array}{lll} \underset{x}{\text{minimize}} & f(x) & \textit{objective} \\ \text{subject to} & c(x) \geq 0 & \textit{constraints} \\ & 0 \leq x_1 \perp x_2 \geq 0 & \textit{complementarity} \end{array} \right.$$

where $x = (x_0, x_1, x_2)$ ... partition of variables and

$$0 \leq x_1 \perp x_2 \geq 0 \quad \Leftrightarrow \quad \text{either } x_{1i} = 0 \text{ or } x_{2i} = 0$$

... equality constraints $h(x) = 0$ are no problem!

EXAMPLE: COMMODITIES FLOW IN NETWORK

Commodities Flow in a Network

- variables are flow on arcs & total flow:

```
var x {ARCS, COMMODITIES} >= 0;
var f {(i,j) in ARCS}
    = sum {k in COMMODITIES} x[i,j,k];
```

- each player (commodity) minimizes time in network

```
minimize time {k in COMMODITIES}:
   sum {(i,j) in ARCS} ( alpha[i,j]*f[i,j]
       + beta[i,j]*(f[i,j]/kappa[i,j]/5)^5);
```

- subject to conservation of flow (supply/demand)

```
subject to conserve {i in NODES, k in COMMODITIES}:
   sum {(i,j) in ARCS} x[i,j,k]
   <= sum{(j,i) in ARCS} x[j,i,k] + b[i,k];
```

Flow of Commodities Through a Network

```
set NODES;                                    # Nodes in network
set ARCS within NODES cross NODES;            # Arcs in network
set COMMODITIES;                              # Commodities

param b {NODES, COMMODITIES} default 0;  # Supply/demand
param alpha {ARCS} >= 0;                      # Linear part
param beta {ARCS} >= 0;                       # Nonlinear part
param kappa {ARCS} >= 0;                      # Nonlinear part

var x {ARCS, COMMODITIES} >= 0;               # Flow on arcs
var f {(i,j) in ARCS}                         # Total flow
    = sum {k in COMMODITIES} x[i,j,k];
```

Flow of Commodities Through a Network

```
minimize time {k in COMMODITIES}:
   sum {(i,j) in ARCS} (alpha[i,j]*f[i,j]
       + beta[i,j]*(f[i,j]/kappa[i,j]/5)^5);

subject to
   conserve {i in NODES, k in COMMODITIES}:
       sum {(i,j) in ARCS} x[i,j,k]
       <= sum{(j,i) in ARCS} x[j,i,k] + b[i,k];

problem player {k in COMMODITIES}:
   time[k],                              # objective
   {i in NODES} conserve[i,k],           # constraints
   {(i,j) in ARCS} x[i,j,k], f;          # variables
```

Nash Game: Complementarity Model

```
subject to
  d_obj {(i,j) in ARCS, k in COMMODITIES}:
    0 <= x[i,j,k] complements
      alpha[i,j] + beta[i,j]*(f[i,j]/kappa[i,j])^4
      + p[i,k] >= p[j,k];

  conserve {i in NODES, k in COMMODITIES}:
    0 <= p[i,k] complements
      sum{(j,i) in ARCS} x[j,i,k] + d[i,k]
      >= sum {(i,j) in ARCS} x[i,j,k];

  flow {(i,j) in ARCS} :
    f[i,j] = sum {k in COMMODITIES} x[i,j,k]
    complements f[i,j];  ### not needed here
```

Stackelberg Game: MPEC

- introduce toll arcs (faster, but carry cost)
- leader chooses tolls to maximize profit ...
  ... or minimize congestion
- followers move optimally (Wardrop's principle)



Stackelberg Game: MPEC

```
set NODES;                                  # Nodes in newtork
set ARCS within NODES cross NODES;          # Arcs in network
set COMMODITIES;                            # Commodities

param b {NODES, COMMODITIES} default 0;     # Supply/demand
param alpha {ARCS} >= 0;                    # Linear part
param beta {ARCS} >= 0;                     # Nonlinear part
param kappa {ARCS} >= 0;                    # Nonlinear part

var t {(i,j) in TOLL}
    >= tl[i,j], <= tu[i,j],                 # Tolls
    := (tl[i,j] + tu[i,j])/2 ;

var x {ARCS, COMMODITIES};                  # Flow on arcs
var p {NODES, COMMODITIES};                 # Multipliers
var f {(i,j) in ARCS};                      # Total flow
```

Stackelberg Game: MPEC

```
maximize revenue:  sum {(i,j) in TOLL} t[i,j]*f[i,j];

subject to
  d_obj {(i,j) in ARCS, k in COMMODITIES}:
    0 <= x[i,j,k] complements
      alpha[i,j] + beta[i,j] * (f[i,j]/kappa[i,j])^4
      + (if (i,j) in TOLL then t[i,j] else 0)
      + p[i,k] >= p[j,k];

  conserve {i in NODES, k in COMMODITIES}:
    0 <= p[i,k] complements
      sum{(j,i) in ARCS} x[j,i,k] + b[i,k]
      >= sum {(i,j) in ARCS} x[i,j,k];

  flow {(i,j) in ARCS}:
    f[i,j] = sum {k in COMMODITIES} x[i,j,k];
```

Solving Large Stackelberg Games

Toll pricing models get large ...

# nodes	# variables	# constraints
5	112	109
25	5000	2500

... depending on arches etc ...

Solving Large Stackelberg Games

Toll pricing models get large ...

# nodes	# variables	# constraints
5	112	109
25	5000	2500

... depending on arches etc ...

Exploit NCP solvers to set-up MPECs:

1. fix leader's variables & solve NCP (Nash)
2. unfix leader's variables & solve MPEC (Stackelberg)

$\Rightarrow$ orders of magnitude faster than MPEC ...

From Nash to Stackelberg with AMPL

```
model wardrop_tax.mod;
data wardrop_tax.dat;

problem NashGame:
   ### variables ... t[i,j] fixed automatically
   {(i,j) in ARCS} f[i,j],
   {(i,j) in ARCS, k in COMMODITIES} x[i,j,k],
   {i in NODES,    k in COMMODITIES} p[i,k],
   ### constraints
   {(i,j) in ARCS, k in COMMODITIES} d_obj[i,j,k],
   {i in NODES,    k in COMMODITIES} conserve[i,k],
   {(i,j) in ARCS}                   flow[i,j];

# ... solve NCP with fixed tolls
option solver pathampl; solve NashGame;
```

From Nash to Stackelberg with AMPL

```
problem Stackelberg:
   revenue,                                  # objective
   ### variables ... added tolls t[i,j]
   {(i,j) in TOLL} t[i,j], {(i,j) in ARCS} f[i,j],
   {(i,j) in ARCS, k in COMMODITIES} x[i,j,k],
   {i in NODES,    k in COMMODITIES} p[i,k],
   ### constraints
   {(i,j) in ARCS, k in COMMODITIES} d_obj[i,j,k],
   {i in NODES,    k in COMMODITIES} conserve[i,k],
   {(i,j) in ARCS}                   flow[i,j];

option mpec_options "compl_ini=1";   # hot start
option solver mpec; solve Stackelberg;

display f; display tl, t, tu;
```

MORE EXAMPLES

More Examples of Stackelberg Games

- CO_2 abatement & transboundary pollution
- interaction of NO_x & electricity markets
- volatility estimation in American option pricing
- transportation network design

MPECs allow us to ...

1. model complex games (heterogeneous players):
   e.g. electricity markets: ISO, generator
2. simulate large & complex games
   ... may need NCP approach first ...
3. extend to design of markets
   e.g. toll-pricing (where to raise tolls)
   $\Rightarrow$ integer variables ... harder problem

FORMULATION OF MPECs AS NLPs

Mathematical Program with Equilibrium Constraints

Mathematical Program with Equilibrium Constraints (MPEC)

$$\left\{ \begin{array}{lll} \underset{x}{\text{minimize}} & f(x) & \textit{objective} \\ \text{subject to} & c(x) \geq 0 & \textit{constraints} \\ & 0 \leq x_1 \perp x_2 \geq 0 & \textit{complementarity} \end{array} \right.$$

where $x = (x_0, x_1, x_2)$ ... partition of variables and

$$0 \leq x_1 \perp x_2 \geq 0 \quad \Leftrightarrow \quad \text{either } x_{1_i} = 0 \text{ or } x_{2_i} = 0$$

... equality constraints $h(x) = 0$ are no problem!

A Nonlinear Programming Approach

Replace equilibrium $0 \leq x_1 \perp x_2 \geq 0$ by $x_1^T x_2 \leq 0$

$\Rightarrow$ standard nonlinear program (NLP)

$$
\text{(NLP)} \left\{ \begin{array}{ll} \underset{x}{\text{minimize}} & f(x) \\ \text{subject to} & c(x) \geq 0 \\ & x_1, \; x_2 \geq 0 \\ & \boxed{X_1 x_2 \leq 0} \end{array} \right.
$$



Advantage: standard (?) NLP; use large-scale solvers ...
Snag: nonlinear program (NLP) violates standard assumptions!

Mangasarian Fromowitz CQ fails

Mangasarian Fromowitz Constraint Qualification at feasible $\hat{x}$:

$$\hat{x}_1 = 0, \ \hat{x}_2 > 0$$

$$\Rightarrow \quad x_1 \geq 0, \quad \text{and} \quad x_2 x_1 \leq 0 \quad \text{active}$$

$$\Rightarrow \quad s_1 > 0, \quad \text{and } \hat{x}_2 s_1 < 0$$



MFCQ is minimalistic stability assumption for NLP

Failure of MFCQ implies:

1. Lagrange multiplier set unbounded

Mangasarian Fromowitz CQ fails

Mangasarian Fromowitz Constraint Qualification at feasible $\hat{x}$:

$$\hat{x}_1 = 0, \ \hat{x}_2 > 0$$

$$\Rightarrow \quad x_1 \geq 0, \quad \text{and} \quad x_2 x_1 \leq 0 \quad \text{active}$$

$$\Rightarrow \quad s_1 > 0, \quad \text{and } \hat{x}_2 s_1 < 0$$



MFCQ is minimalistic stability assumption for NLP

Failure of MFCQ implies:

1. Lagrange multiplier set unbounded
2. Constraint gradients linearly dependent

Mangasarian Fromowitz CQ fails

Mangasarian Fromowitz Constraint Qualification at feasible $\hat{x}$:

$$\hat{x}_1 = 0, \ \hat{x}_2 > 0$$

$$\Rightarrow \quad x_1 \geq 0, \quad \text{and} \quad x_2 x_1 \leq 0 \quad \text{active}$$

$$\Rightarrow \quad s_1 > 0, \quad \text{and} \ \hat{x}_2 s_1 < 0$$



MFCQ is minimalistic stability assumption for NLP

Failure of MFCQ implies:

1. Lagrange multiplier set unbounded
2. Constraint gradients linearly dependent
3. Central path does not exist

Mangasarian Fromowitz CQ fails

Early (1980s) numerical experience disappointing:

- failure on 60% of problems!!!

Popular conclusion:

- NLP approach is "inherently unstable"
- must expect "failure in presence of round-off errors"
- arbitrary small perturbation $\epsilon > 0$: $\Rightarrow x_1^T x_2 \leq -\epsilon \Rightarrow$ **infeasible**

Mangasarian Fromowitz CQ fails

Early (1980s) numerical experience disappointing:

- failure on 60% of problems!!!

Popular conclusion:

- NLP approach is "inherently unstable"
- must expect "failure in presence of round-off errors"
- arbitrary small perturbation $\epsilon > 0$: $\Rightarrow x_1^T x_2 \leq -\epsilon \Rightarrow$ **infeasible**

Homer Simpson Approach to Numerical Analysis ...

- ... who perturbs structural zeros ???
- ... huge advances in NLP technology in 90s ...

CONVERGENCE OF NLP APPROACH

MPECs as NLPs

Mathematical Program with Equilibrium Constraints (MPEC)

$$\left\{ \begin{array}{lll} \underset{x}{\text{minimize}} & f(x) & \textit{objective} \\ \text{subject to} & c(x) \geq 0 & \textit{constraints} \\ & 0 \leq x_1 \perp x_2 \geq 0 & \textit{complementarity} \end{array} \right.$$

... formulated as nonlinear program (NLP):

$$\left\{ \begin{array}{ll} \underset{x}{\text{minimize}} & f(x) \\ \text{subject to} & c(x) \geq 0 \\ & x_1, \ x_2 \geq 0 \\ & \boxed{X_1 x_2 \leq 0} \end{array} \right.$$

Stationarity & Bounded Multipliers

Example $x^* = (0,1)$:

$$\left\{ \begin{array}{ll} \min\limits_x & \frac{1}{2}(x_1 - 1)^2 + (x_2 - 1)^2 \\ \text{s.t.} & x_1, x_2 \geq 0, \; x_1 x_2 \leq 0 \end{array} \right.$$

first order conditions:

$$\begin{pmatrix} -1 \\ 0 \end{pmatrix} = \begin{pmatrix} \nu_1 \\ 0 \end{pmatrix} - \begin{pmatrix} \xi \\ 0 \end{pmatrix}$$

ν_1 multiplier of $x_1 \geq 0$; ξ multiplier of $x_1 x_2 \leq 0$.
Equivalent NLP ($x_1 x_2 \leq 0$) violates MFCQ
$\Rightarrow$ unbounded multipliers





multipliers form a ray $\Rightarrow \exists$ bounded multipliers

Optimality Conditions for MPECs

Optimality conditions of equivalent NLP: $\exists \lambda^*, \nu_1^*, \nu_2^*, \xi^* \geq 0$

$$\nabla f(x^*) - \nabla c(x^*)^T \lambda^* - \begin{pmatrix} 0 \\ \nu_1^* - X_2^* \xi^* \\ \nu_2^* - X_1^* \xi^* \end{pmatrix} = 0 \quad 1^{st} \text{ order}$$

$$c(x^*) \geq 0, \; x_1^* \geq 0, \; x_2^* \geq 0 \text{ and } X_1^* x_2^* \leq 0 \quad \text{primal feas.}$$

$$c(x^*)^T \lambda \; = \; {x_1^*}^T \nu_1^* \; = \; {x_2^*}^T \nu_2^* = 0 \quad \text{compl. slack.}$$

... $\xi > 0$ allows $\nu_1^* - X_2^* \xi^* < 0$ ... equality constraint $x_1 = 0$

Multipliers bounded if $\|\xi^*\| < \infty$

Convergence of SQP Methods

SQP: quadratic/linear approximation of NLP [Todd's talk]

Theorem: SQP converges quadratically!

Proof: ... paraphrased from 7 pages ...

1. if ${x_1^{(k)}}^T x_2^{(k)} = 0$ then ${x_1^{(k+1)}}^T x_2^{(k+1)} = 0$
   $\Rightarrow$ standard SQP convergence proof
2. if ${x_1^{(k)}}^T x_2^{(k)} > 0$ then ...
   ... QP solver always picks non-singular basis
   $\Rightarrow$ standard SQP convergence proof

Do Slacks Matter ???

Consider general equilibrium $0 \le G(x) \perp H(x) \ge 0$:

Do Slacks Matter ???

Consider general equilibrium $0 \leq G(x) \perp H(x) \geq 0$:
MPEC with nonlinear complementarity condition
"nice" stationary points at $(0,2)$ and $(2,0)$

$$\left\{ \begin{array}{ll} \underset{x}{\text{minimize}} & -x_1 - \frac{1}{2}x_2 \\ \text{subject to} & x_1 + x_2 \leq 2 \\ & 0 \leq x_1^2 - x_1 \perp x_2 \geq 0 \qquad \text{nonlinear} \end{array} \right.$$

$x_0 = (-\epsilon, t)^T$ for $t \geq 0$:

- $x_k \to (0,t)^T$ quadratically
- $\xi \to \infty$ weird !!!
- active set singular in limit

$\Rightarrow$ SQP gets stuck



DETOUR:
INTERIOR POINT METHODS

Interior Point Methods (IPM)

General NLP

$$\underset{x}{\text{minimize}} \; f(x) \quad \text{subject to } c(x) = 0 \quad \& \quad x \geq 0$$

Perturbed $\mu > 0$ optimality conditions $(x, z \geq 0)$

$$F_\mu(x, y, z) = \left\{ \begin{array}{r} \nabla f(x) - \nabla c(x)^T y - z \\ c(x) \\ Xz - \mu e \end{array} \right\} = 0$$

- Primal-dual formulation
- Central path $\{x(\mu), y(\mu), z(\mu) \; : \; \mu > 0\}$
- Apply Newton's method for sequence $\mu \searrow 0$

Interior Point Methods (IPM)

Newton's method applied to primal-dual system gives ...

$$\begin{bmatrix} \nabla^2 \mathcal{L}_k & -A_k & -I \\ A_k^T & 0 & 0 \\ Z_k & 0 & X_k \end{bmatrix} \begin{pmatrix} \Delta x \\ \Delta y \\ \Delta z \end{pmatrix} = -F_\mu(x_k, y_k, z_k)$$

where $A_k = \nabla c(x_k)^T$, X_k diagonal matrix of x_k.

Solvers: `LOQO`, `KNITRO`, `IPOPT` ... different factorization

Interior Point Methods (IPM)

$$\underset{x}{\text{minimize}}\ f(x) \quad \text{subject to } c(x) = 0 \quad \& \quad x \geq 0$$

Related to barrier methods for handling inequalities

$$\begin{cases} \underset{x}{\text{minimize}} & f(x) - \mu \sum \log(x_i) \\ \text{subject to} & c(x) = 0 \end{cases}$$



$$\text{minimize } x_1^2 + x_2^2 \quad \text{subject to} \quad x_1 + x_2^2 \geq 1$$

Interior Point Methods (IPM)

$$\underset{x}{\text{minimize}} \; f(x) \quad \text{subject to } c(x) = 0 \quad \& \quad x \geq 0$$

Related to barrier methods for handling inequalities

$$\begin{cases} \underset{x}{\text{minimize}} & f(x) - \mu \sum \log(x_i) \\ \text{subject to} & c(x) = 0 \end{cases}$$



$$\text{minimize } x_1^2 + x_2^2 \quad \text{subject to} \quad x_1 + x_2^2 \geq 1$$

BACK TO MPECs

Convergence of Interior Point Methods

Key idea: MPEC-stationarity $\Leftrightarrow$ 1^{st} order conditions of NLP

Approaches:

Convergence of Interior Point Methods

Key idea: MPEC-stationarity $\Leftrightarrow$ 1^{st} order conditions of NLP

Approaches:

1. Penalize equilibrium minimize $f(x) + \pi x_1^T x_2$
   - Well behaved smooth problem: constraints satisfy MFCQ
   - $\pi^* < \infty$ and $X_1 x_2 \leq 0 \Rightarrow$ strongly-stationary
   - How to adjust π? ... during barrier solve!

Convergence of Interior Point Methods

Key idea: MPEC-stationarity $\Leftrightarrow$ 1^{st} order conditions of NLP

Approaches:

1. Penalize equilibrium minimize $f(x) + \pi x_1^T x_2$
   - Well behaved smooth problem: constraints satisfy MFCQ
   - $\pi^* < \infty$ and $X_1 x_2 \leq 0 \Rightarrow$ strongly-stationary
   - How to adjust π? ... during barrier solve!
2. Relax equilibrium $X_1 x_2 \leq \tau$ & $x_1, x_2 \geq -\delta$
   - Well behaved smooth problem
   - Adjust $\tau_i \searrow 0$ or $\delta_i \searrow 0$ ... not both

Convergence of Interior Point Methods

Key idea: MPEC-stationarity $\Leftrightarrow$ 1^{st} order conditions of NLP

Approaches:

1. Penalize equilibrium minimize $f(x) + \pi x_1^T x_2$
   - Well behaved smooth problem: constraints satisfy MFCQ
   - $\pi^* < \infty$ and $X_1 x_2 \leq 0 \Rightarrow$ strongly-stationary
   - How to adjust π? ... during barrier solve!
2. Relax equilibrium $X_1 x_2 \leq \tau$ & $x_1, x_2 \geq -\delta$
   - Well behaved smooth problem
   - Adjust $\tau_i \searrow 0$ or $\delta_i \searrow 0$ ... not both

Aim: NLP solver with small modification works for MPECs

Interior Point Method with Two Sided Relaxation

- MPECs have no strict interior
- Relaxation $X_1 x_2 \leq \tau \searrow 0 \Rightarrow$ interior $\rightarrow 0$

$\Rightarrow$ relax $X_1 x_2 \leq \tau$
and $x_1 \geq -\delta$, $x_2 \geq -\delta$
... adjust τ, δ as $\mu \rightarrow 0$



Theorem: In limit $\tau_i \rightarrow 0$ or $\delta_i \rightarrow 0$ but not both
$\Rightarrow$ relaxed problem has non-empty interior in limit

MPEC multiplier $\mu_i < 0 \Rightarrow$ reduce $\tau_i \searrow 0$ ...

Limitations of NLP Approach

- formulation of followers as NCP
  $\Rightarrow$ questionable for nonconvex games
- MPEC-stationarity not necessarily no descent
- multipliers $\xi \to \infty$ ... not a solution???

... NLP approach remains most versatile MPEC solver
... allows simulation of heterogeneous leader-follower games

MULTI-LEADER-FOLLOWER GAMES

Multi-Leader-Follower Games

Non-cooperative equilibrium between >2 dominant producers

$$x_i^*, y \in \left\{ \begin{array}{ll} \underset{x_i \geq 0, y}{\text{argmin}} & f_i(x, y) \\ \text{s.t.} & h(x_i, y) = 0 \\ & 0 \leq y_1 \perp y_2 \geq 0 \end{array} \right.$$



Complementarity: optimality conditions of leaders
$\Rightarrow$ equilibrium problem with equilibrium constrains (EPEC)

Leaders' MPECs violate Mangasarian-Fromowitz CQ
... are EPECs harder than MPECs ???

... recall MPEC strong-stationarity $\Rightarrow \exists$ bounded multipliers

A Practical Nonlinear Approach to EPECs

Initial experience: no feasible solution to MLF game

Equilibrium may not exist:

$$\begin{cases} \text{minimize} & \begin{pmatrix} \frac{1}{2}x_1 + z \\ -\frac{1}{2}x_2 + z \end{pmatrix} \\ \text{subject to} & z - 1 + x_1 - x_2 - \lambda = 0 \\ & 0 \le z \perp \lambda \ge 0 \end{cases}$$



$\Rightarrow z(x_1, x_2) = \max(0, 1 - x_1 - x_2)$

A Practical Nonlinear Approach to EPECs

Initial experience: no feasible solution to MLF game

Equilibrium may not exist:

$$\begin{cases} \text{minimize} & \begin{pmatrix} \frac{1}{2}x_1 + z \\ -\frac{1}{2}x_2 + z \end{pmatrix} \\ \text{subject to} & z - 1 + x_1 - x_2 - \lambda = 0 \\ & 0 \le z \perp \lambda \ge 0 \end{cases}$$



$\Rightarrow z(x_1, x_2) = \max(0, 1 - x_1 - x_2)$
... suggest convexification approach to get feasible ...

Gauss-Seidel Iteration for EPECs

```
set I := 1..2;

var x{I};          # ... leader variables
var s;             # ... follower variables
var y;

# ... leader's objective
minimize leader{i in I}: (x[i]+1)^2;

subject to
  slack: s = x[1] + x[2] + y;
  compl: 0 <= s   complements   y >= 0;
```

Gauss-Seidel Iteration for EPECs

```
param x0{I} default 0;                        # iter k
param s0{I} default 0;
param y0{I} default 0;
param x1{I}; param s1{I}; param y1{I};  # iter k+1

param Tol >= 0, default 1E-4;               # tolerance
param MaxIt >= 0, integer, default 20;  # max iter
param Err default 0;

let{i in I} x[i] := x0[i];                   # init vars
let s := s0[1]; let y := y0[1];

problem Stackelberg{i in I}:                # leader i
        leader[i],                          # objective
        x[i], y, s,                         # variables
        slack, compl;                       # constraints
```

Gauss-Seidel Iteration for EPECs

```
for {k in 1..MaxIt}{
   for {i in I}{
      solve Stackelberg[i];
      let x1[i] := x[i];
      let s1[i] := s;    let y1[i] := y;
   }; # end for

   let Err := sqrt( sum{i in I}( (x0[i] - x1[i])^2
                               + (y0[i] - y1[i])^2
                               + (s0[i] - s1[i])^2 ));
   if (Err <= Tol) then break;

   let{i in I} x0[i]:=x1[i]; let{i in I} s0[i]:=s1[i];
   let{i in I} y0[i]:=y1[i];
}; # end for
```

Complementarity System Approach

Non-cooperative Nash equilibrium between leaders $(\nabla_i = \nabla_{(x_i, y)})$ ... for all leaders $i = 1, \ldots, L$:

$$\nabla_i f_i(x,y) - \nabla_i h(x_i, y)\mu_i - \begin{pmatrix} \sigma_i \\ 0 \\ \nu_{i1} - \xi_i y_2 \\ \nu_{i2} - \xi_i y_1 \end{pmatrix} = 0$$

$$h(x_i, y) = 0$$

$$0 \le x_i \perp \sigma_i \ge 0 \quad , \quad 0 \ge y_1^T y_2 \perp \xi_i \ge 0$$

$$0 \le y_1 \perp \nu_{i1} \ge 0 \quad , \quad 0 \le y_2 \perp \nu_{i2} \ge 0$$

Non-square complementarity problem

... OK for nonlinear optimizers

... note each leader has own multipliers ξ_i, ν_i

... price-consistent formulation $\xi_i = \xi \; \forall i \Rightarrow$ square NCP

Complementarity System Approach

```
var lambda{I}; var sigma{I};  # ... multipliers
var nu{I};     var xi{I};

subject to
  KKTx{i in I}: 2*(x[i]+1) + lambda[i] = 0;
  KKTy{i in I}:   lambda[i] - nu[i]     + xi[i]*s = 0;
  KKTs{i in I}: - lambda[i] - sigma[i] + xi[i]*y = 0;

  slack{i in I}: s = x[1]+x[2]+y complements lambda[i];

  bnds{i in I}:  0 <= s    complements  sigma[i] >= 0;
  bndy{i in I}:  0 <= y    complements  nu[i]    >= 0;
  compl{i in I}: 0 <= -y*s complements  xi[i]    >= 0;
```

Solution of Multi-Leader Follower Games

- MPECs are nonconvex
  $\Rightarrow$ first-order conditions are not sufficient
  $\Rightarrow$ NCP approach may fail
- Differentiation is not easy
  $\Rightarrow$ mistakes in NCP formulation
- MPECs are already difficult
  $\Rightarrow$ Gauss-Seidel often fails
- penalized approach works best at present

CONCLUSIONS

Conclusion & Outlook

Conclusions

- complex leader follower games are MPECs/EPECs
- numerical treatment of large problems possible

$\Rightarrow$ can simulate complex games numerically

Conclusion & Outlook

Conclusions

- complex leader follower games are MPECs/EPECs
- numerical treatment of large problems possible

$\Rightarrow$ can simulate complex games numerically

Open Questions

- Interpretation of NLP failures?
- Can we avoid "spurious-stationary" points?
- How to get global minimizers for some applications?

"Homework"

1. Formulate `hakonsen.mod` as an NLP;
   replacing $0 \leq x_1 \perp x_2 \geq 0$ by
   1.1 $\min(x_1, x_2) \leq 0$ and $x_1, x_2 \geq 0$
   1.2 $x_1^T x_2 \leq 0$ and $x_1, x_2 \geq 0$

   ... and solve the model using `snopt`, `loqo` et al.
2. Solve `wardop.mod/wardrop-01.dat`:
   2.1 starting from default
   2.2 starting from point given in `wardrop-01.start`

   What is the effect on the solvers? Compare IPM to ASM!

Models: `www.mcs.anl.gov/~leyffer/ice05/`